

# KPL Syntax

*Harry H. Porter III  
Portland State University*

[HHPorter3@gmail.com](mailto:HHPorter3@gmail.com)

*14 December 2023*

This document gives syntax of the KPL programming language.  
KPL is the programming language of the Blitz-64 Computer  
System.

# Table of Contents

<b>Notation Used in the Grammar</b>	<b>3</b>
<b>A Context Free Grammar of KPL</b>	<b>5</b>
<b>Expressions</b>	<b>9</b>
Precedence of Operators	10
<b>Lexical Matters</b>	<b>12</b>
Comments and White Space	12
Identifiers	12
Integers	13
Floating Point Constants	13
Character Constants	13
String Constants	14
Escape Sequences	15
Keyword List	16
<b>About This Document</b>	<b>17</b>
Document Revision History	17
Permission to Copy	17

# Notation Used in the Grammar

This document provides a Context-Free Grammar (CFG) for the KPL language<sup>1</sup>. This grammar is meant to be exactly identical to the grammar the appendix of the document

*“An Introduction to KPL: A Kernel Programming Language”*

To make the grammar easier to read and understand, we use an extended CFG notation, which is described here.

*Non-terminal Symbols are shown like this:*

HeaderFile    Type    Expr    Statement    etc...

*Terminal Symbols:*

*Keywords are shown in boldface, like this:*

**if**    **while**    **int**    **endWhile**    etc...

*The following lexical tokens appear in the grammar:*

	<i>Examples</i>	
INTEGER	<b>42</b>	<b>0x1234ABCD</b>
DOUBLE	<b>3.1415</b>	<b>6.022e23</b>
CHAR	<b>'a'</b>	<b>'\n'</b>
STRING	<b>"hello"</b>	<b>"\t\n"</b>
ID	<b>myName</b>	<b>MAX_SIZE</b>
OPERATOR	<b>&lt;=</b> <b>&lt;</b> <b>&gt;</b> <b>&gt;=</b> <b>!=</b> <b>+</b> <b>-</b> <b>*</b>	<i>etc...</i>

*Punctuation Symbols*

*The following characters are particularly important in KPL's grammar:*

{ } [ ] | : , . = ( ) ;

*Of these, the following punctuation symbols conflict with grammar meta-symbols:*

{ } [ ] |

*When used as grammar meta-symbols, they are shown without quotes:*

{ } [ ] |

*When used as terminals, i.e., when meant literally, they are quoted:*

'{ '}' '[' ']' '|'

*The remaining punctuation symbols are only used as terminals and are not quoted:*

: , . = ( ) ;

---

<sup>1</sup> Technically, the KPL grammar is “LL(k)” and in most cases can be parsed with only a single token look-ahead, making it much easier for the human than “LR(k)” grammars.

*Comments are not included in this grammar. There are two forms of commenting:*

*-- through end-of-line*  
*/\* through \*/*

*Meta-Symbols, which are used in describing the grammar:*

*Grammar rules: -->*

*Example:*

Type --> **int**

*Rules with alternatives are shown like this:*

*Example:*

Statement --> IfStmt | AssignStmt

*Example:*

Statement --> IfStmt  
          --> AssignStmt

*Optional material: [ ]*

*Example:*

VarDecl --> Decl [ = Expr2 ]

*Repetition of zero-or-more: { }*

*Example:*

StmtList --> { Statement }

*Repetition of one-or-more occurrences: { }+*

*Example:*

VarDecls --> **var** { VarDecl }+

# A Context Free Grammar of KPL

```
HeaderFile      --> header ID
                 [ Uses ]
                 { Constants |
                   Errors |
                   VarDecls |
                   Enum |
                   TypeDefs |
                   FunctionProtos |
                   Interface |
                   Class }
                 endHeader
CodeFile        --> code ID
                 { Constants |
                   Errors |
                   VarDecls |
                   Enum |
                   TypeDefs |
                   Function |
                   Interface |
                   Class |
                   Behavior }
                 endcode
Interface       --> interface ID [ TypeParms ]
                 [ extends TypeList ]
                 [ messages { MethProto }+ ]
                 endInterface
Class           --> class ID [ TypeParms ]
                 [ implements TypeList ]
                 [ superclass NamedType ]
                 [ fields { Decl }+ ]
                 [ methods { MethProto }+ ]
                 endClass
Behavior        --> behavior ID
                 { Method }
                 endBehavior
Uses            --> uses OtherPackage { , OtherPackage }
OtherPackage    --> ID      [ renaming Rename { , Rename } ]
                --> STRING [ renaming Rename { , Rename } ]
Rename          --> ID to ID
TypeParms       --> '[' ID : Type { , ID : Type } ']'
Constants       --> const { ID = Expr }+
Decl            --> ID { , ID } : Type
VarDecl         --> Decl [ = Expr2 ]
VarDecls        --> var { VarDecl }+
Errors          --> errors { ID ParmList }+
TypeDefs        --> type { ID = Type }+
Enum            --> enum ID [ = Expr ] { , ID }
IdList          --> ID { , ID }
ArgList         --> ( )
                --> ( Expr { , Expr } )
```

```

ParmList      --> ( )
              --> ( Decl { , Decl } )
FunctionProtos --> functions { FunProto }+
FunProto     --> [ external ] ID ParmList [ returns Type ] [ StackUsage ]
Function     --> function ID ParmList [ returns Type ] [ StackUsage ]
              [ VarDecls ]
              StmtList
              endFunction
StackUsage   --> '[' Max_Stack_Usage = Expr ']'
NamelessFunction --> function ParmList [ returns Type ]
              [ VarDecls ]
              StmtList
              endFunction
MethProto    --> ID ParmList [ returns Type ] [ StackUsage ]
              --> infix OPERATOR ( ID : Type ) returns Type
              --> prefix OPERATOR ( ) returns Type
              --> { ID : ( ID : Type ) }+ [ returns Type ]
Method       --> method MethProto
              [ VarDecls ]
              StmtList
              endMethod
StmtList     --> { Statement }
Statement    --> if Expr StmtList
              { elseif Expr StmtList }
              [ else StmtList ]
              endif
              --> LValue = Expr
              --> LValue += Expr
              --> LValue -= Expr
              --> ID ArgList
              --> Expr { ID : Expr }+
              --> Expr . ID ArgList
              --> while Expr
              StmtList
              endWhile
              --> do
              StmtList
              until Expr
              --> break
              --> continue
              --> return [ Expr ]
              --> for LValue = Expr to Expr [ by Expr ]
              StmtList
              endFor
              --> for ( StmtList ; [ Expr ] ; StmtList )
              StmtList
              endFor
              --> switch Expr
              { case Expr : StmtList }
              [ default : StmtList ]
              endSwitch

```

	<pre> --&gt; <b>switchOnClass</b> Expr       { <b>case</b> Expr : StmtList }       [ <b>default</b> : StmtList ]       <b>endSwitchOnClass</b> --&gt; <b>try</b> StmtList       { <b>catch</b> ID ParmList : StmtList }+       <b>endTry</b> --&gt; <b>throw</b> ID ArgList --&gt; <b>free</b> Expr --&gt; <b>debug</b> [ STRING ] --&gt; <b>printf</b> ( [ ID , ] STRING { , Expr } ) --&gt; <b>sprintf</b> ( ID , STRING { , Expr } ) --&gt; <b>initializeArray</b> ( Expr ) --&gt; <b>setArraySize</b> ( Expr , Expr ) --&gt; <b>byte</b> --&gt; <b>halfword</b> --&gt; <b>word</b> --&gt; <b>int</b> --&gt; <b>double</b> --&gt; <b>bool</b> --&gt; <b>void</b> --&gt; <b>typeOfNull</b> --&gt; <b>anyType</b> --&gt; <b>ptr to</b> Type --&gt; <b>struct</b> { Decl }+ <b>endStruct</b> --&gt; <b>union</b> { Decl }+ <b>endUnion</b> --&gt; <b>array</b> [ '[' Dimension { , Dimension } ']' ] <b>of</b> Type --&gt; <b>function</b> ( [ Type { , Type } ] )       [ <b>returns</b> Type ] [ StackUsage ] --&gt; NamedType NamedType --&gt; ID [ '[' Type { , Type } ']' ] TypeList --&gt; NamedType { , NamedType } Dimension --&gt; *   Expr Constructor --&gt; Type ClassStructInit --&gt; Type ArrayInit --&gt; Type ClassStructInit --&gt; ID '{' ID = Expr { , ID = Expr } '}' ArrayInit --&gt; ID '{' [ Expr <b>of</b> ] Expr       { , [ Expr <b>of</b> ] Expr } '}'  LValue --&gt; Expr Expr --&gt; Expr2 { ID : Expr2 } Expr2 --&gt; Expr3 { OPERATOR Expr3 } Expr3 --&gt; Expr5 { ' ' Expr5 } Expr5 --&gt; Expr6 { '&amp;&amp;' Expr6 } Expr6 --&gt; Expr7 { ' ' Expr7 } Expr7 --&gt; Expr8 { '^' Expr8 } Expr8 --&gt; Expr9 { '&amp;' Expr9 } Expr9 --&gt; Expr10 { '==' Expr10         '!=' Expr10 } Expr10 --&gt; Expr11 { '&lt;' Expr11         '&lt;=' Expr11         '&gt;' Expr11         '&gt;=' Expr11 } </pre>
--	---

```
Expr11      --> Expr12 {   << Expr12
                    |   >> Expr12
                    |  <<< Expr12
                    |  >>> Expr12 }
Expr12      --> Expr13 {   + Expr13
                    |   - Expr13 }
Expr13      --> Expr15 {   * Expr15
                    |   / Expr15
                    |   % Expr15 }
Expr15      --> OPERATOR Expr15
Expr16      --> Expr16
Expr16      --> Expr17 {   . ID ArgList
                    |   . ID
                    |   '[' Expr { , Expr } ']' }
Expr17      --> ( Expr )
Expr17      --> null
Expr17      --> true
Expr17      --> false
Expr17      --> self
Expr17      --> super
Expr17      --> INTEGER
Expr17      --> DOUBLE
Expr17      --> CHAR
Expr17      --> STRING
Expr17      --> NamelessFunction
Expr17      --> ID
Expr17      --> ID ArgList
Expr17      --> new Constructor
Expr17      --> alloc Constructor
Expr17      --> sizeof ( Type )
Expr17      --> asPtrTo ( Expr , Type )
Expr17      --> asInteger ( Expr )
Expr17      --> arraySize ( Expr )
Expr17      --> arrayMaxSize ( Expr )
Expr17      --> isInstanceOf ( Expr , Type )
Expr17      --> isKindOf ( Expr , Type )
```



# Expressions

Here is simplified grammar for expressions. This rule ignores:

- Precedence
- Associativity

```
Expr --> Expr BinaryOperator Expr
      --> UnaryOperator Expr
      --> Expr . ID ( ...Arguments... )
      --> Expr . ID
      --> Expr '[' Expr { , Expr } ']'
      --> * Expr
      --> & Expr
      --> null | true | false | nan | inf | self | super
      --> ID
      --> INTEGER
      --> DOUBLE
      --> CHAR
      --> STRING
      --> function ( ...Arguments... ) ... endFunction
      --> new Type [ '{' ...Initialization... '}' ]
      --> alloc Type [ '{' ...Initialization... '}' ]
      --> sizeof ( Type )
      --> asPtrTo ( Expr , Type )
      --> asInteger ( Expr )
      --> arraySize ( Expr )
      --> arrayMaxSize ( Expr )
      --> isInstanceOf ( Expr , Type )
      --> isKindOf ( Expr , Type )
      --> ID ( ArgList )
      --> ( Expr )

BinaryOperator --> + | - | * | / | % | >> | << | >>> | <<< |
                < | > | <= | >= | == | != | & | '|' | ^ |
                && | '||' | ...user defined infix operators...


UnaryOperator --> ! | + | - | ...user defined prefix operators...
```

## Precedence of Operators


The formal syntax of KPL imposes the following precedences on these expression operators. Each operator within a group is at the same precedence level and is parsed with left associativity.

This is the same as in C, C++, and Java.

### *Lowest Precedence*



	All keyword messages, e.g., <code>x at:y put:z</code>
	All infix operators not mentioned below
<code>  </code>	Short-circuit for <b>bool</b> operands
<code>&amp;&amp;</code>	Short-circuit for <b>bool</b> operands
<code> </code>	Bitwise OR for <b>int</b> operands
<code>^</code>	Bitwise XOR for <b>int</b> operands
<code>&amp;</code>	Bitwise AND for <b>int</b> operands
<code>==</code> <code>!=</code>	Can compare basic types, pointers, and objects, but not structs, unions or arrays
<code>&lt;</code> <code>&lt;=</code> <code>&gt;</code> <code>&gt;=</code>	Can compare <b>byte</b> , <b>halfword</b> , <b>word</b> , <b>int</b> , <b>double</b> , and <b>ptr</b> operands
<code>&lt;&lt;</code> <code>&gt;&gt;</code> <code>&lt;&lt;&lt;</code> <code>&gt;&gt;&gt;</code>	Shift int operand left logical Shift int operand right logical Shift int operand left arithmetic Shift int operand right arithmetic
<code>+</code> <code>-</code>	Can also add <b>ptr+int</b> Can also subtract <b>ptr-int</b> and <b>ptr-ptr</b>
<code>*</code> <code>/</code> <code>%</code>	Modulo operator for <b>ints</b>
Prefix <code>-</code> Prefix <code>+</code> Prefix <code>!</code> Prefix <code>*</code> Prefix <code>&amp;</code> All other prefix methods	For <b>int</b> and <b>double</b> operands For <b>int</b> and <b>double</b> operands (nop) For <b>int</b> and <b>bool</b> operands Pointer dereference Address-of



.	Message Sending: <code>x.foo(y,z)</code>
.	Field Accessing: <code>x.name</code>
[ ]	Array Accessing: <code>a[i,j]</code>
<hr/>	
( )	Parenthesized expressions: <code>x*(y+z)</code>
constants	e.g., <code>123</code> , <code>"hello"</code> , <code>34.998e-23</code>
keywords	e.g., <b>true</b> , <b>false</b> , <b>null</b> , <b>self</b> , <b>super</b>
nameless funct	e.g., <b>function</b> (...) ... <b>endFunction</b>
variables	e.g., <code>x</code>
function call	e.g., <code>foo(4)</code>
built-ins	e.g., <b>forceToDouble</b> (4)
function	e.g., <b>function</b> (...) ... <b>endFunction</b>
<b>new</b>	e.g., <b>new</b> Person { name="smith" }
<b>alloc</b>	e.g., <b>alloc</b> Person { name="smith" }
<b>sizeof</b>	e.g., <b>sizeof</b> (Person) ... in bytes
<b>asPtrTo</b>	e.g., <b>asPtrTo</b> (i, <b>double</b> )
<b>asInteger</b>	e.g., <b>asInteger</b> (ptr)
<b>arraySize</b>	e.g., <b>arraySize</b> (array/arrayPtr)
<b>arrayMaxSize</b>	e.g., <b>arrayMaxSize</b> (array/arrayPtr)
<b>isInstanceOf</b>	e.g., <b>isInstanceOf</b> (p, <i>ClassName</i> )
<b>isKindOf</b>	e.g., <b>isKindOf</b> (p, <i>ClassOrInterfaceName</i> )

---

*Highest Precedence*

# Lexical Matters

Greater detail about the lexical tokens is given in the KPL Reference Manual. Here is a summary.

## Comments and White Space

KPL supports two comments styles.

First, a comment may begin with `/*` and end with `*/`.

Second, everything after two hyphens through end-of-line is a comment.

```
x = y + 2    -- Adjust y a little
```

Both styles may be nested.

```
/* Disable this code...
  x = y + 2    /* Adjust y a little */
*/

-- Disable this code...
--  x = y + 2    -- Adjust y a little
```

White space is defined as a sequence of one or more of:

- Space
- Tab
- Newline

## Identifiers

An ID is a sequence of letters, digits, and underscores. It must begin with a letter. Only ASCII characters are allowed. Case is significant.

## Integers

An INTEGER can be expressed either in decimal or in hex:

```
12345
0x01b5f3b
```

The underscore can be used as a separator to increase readability. It is ignored:

```
12_345
0x01b_5f3b
```

## Floating Point Constants

A DOUBLE number must contain either a decimal point or the “e” for the exponent.

```
123.0
123e-45
```

For both INTEGERS and DOUBLES, a leading minus/negative sign “-” will be parsed as a separate token and used to form an expression, such as -(1). The compiler will evaluate such expressions at compile-time, so effectively any INTEGER or DOUBLE may be negated.

```
-1      -- Preferred
-(1)    -- Equivalent
```

The underscore can be used as a separator to increase readability. It is ignored:

```
1.000_000_007
```

## Character Constants

A CHAR constant is enclosed in single quotes. Any Unicode character may be included or an escape sequence may be used.

```
'A'
'€'      -- Unicode
'\n'    -- Escape sequence
```

### String Constants

A string constant consists of a sequence of zero or more characters enclosed in double quotes:

```
"Hello, world\n"
```

A string constant is represented as an array of bytes. The UTF-8 encoding scheme is used to represent the string, which may contain arbitrary Unicode characters.

```
" $\pi \approx 3.14$ "
"\U0001d70b \u2248 3.14"  -- Equivalent, with codepoints in hex
"\xf0\x9d\x9c\x8b \xe2\x89\x88 3.14"  -- Equivalent, in UTF-8
```

## Escape Sequences

Here are the escape sequences that may be used in character and string constants.

	Hex	Decimal		ASCII Code	Name
	=====	=====		=====	=====
<code>\0</code>	00	0	control-@	NUL	null
<code>\a</code>	07	7	control-G	BEL	alert
<code>\b</code>	08	8	control-H	BS	backspace
<code>\t</code>	09	9	control-I	HT	tab
<code>\n</code>	0A	10	control-J	NL/LF	newline/linefeed
<code>\v</code>	0B	11	control-K	VT	vertical tab
<code>\f</code>	0C	12	control-L	FF	form feed
<code>\r</code>	0D	13	control-M	CR	return
<code>\e</code>	1B	27	control-[	ESC	escape
<code>\d</code>	7f	127		DEL	delete
<code>\"</code>	22	34		"	double quote
<code>\'</code>	27	39		'	single quote
<code>\\</code>	5C	92		\	backslash
<code>\xHH</code>	HH		< any hex value >		

## Keyword List

Here are the keywords used in the KPL grammar. The built-in function names are not included.

<b>alloc</b>	<b>endMethod</b>	<b>nan</b>
<b>anyType</b>	<b>endStruct</b>	<b>new</b>
<b>array</b>	<b>endSwitch</b>	<b>null</b>
<b>arrayMaxSize</b>	<b>endSwitchOnClass</b>	<b>of</b>
<b>arraySize</b>	<b>endTry</b>	<b>prefix</b>
<b>asInteger</b>	<b>endUnion</b>	<b>printf</b>
<b>asPtrTo</b>	<b>endWhile</b>	<b>ptr</b>
<b>behavior</b>	<b>enum</b>	<b>renaming</b>
<b>bool</b>	<b>errors</b>	<b>return</b>
<b>break</b>	<b>extends</b>	<b>returns</b>
<b>by</b>	<b>external</b>	<b>self</b>
<b>byte</b>	<b>false</b>	<b>setArraySize</b>
<b>case</b>	<b>fields</b>	<b>sizeof</b>
<b>catch</b>	<b>for</b>	<b>sprintf</b>
<b>class</b>	<b>free</b>	<b>struct</b>
<b>code</b>	<b>function</b>	<b>super</b>
<b>const</b>	<b>functions</b>	<b>superclass</b>
<b>continue</b>	<b>halfword</b>	<b>switch</b>
<b>debug</b>	<b>header</b>	<b>switchOnClass</b>
<b>default</b>	<b>if</b>	<b>throw</b>
<b>do</b>	<b>implements</b>	<b>to</b>
<b>double</b>	<b>inf</b>	<b>true</b>
<b>else</b>	<b>infix</b>	<b>try</b>
<b>elseif</b>	<b>initializeArray</b>	<b>type</b>
<b>endBehavior</b>	<b>int</b>	<b>typeOfNull</b>
<b>endClass</b>	<b>interface</b>	<b>union</b>
<b>endcode</b>	<b>isInstanceOf</b>	<b>until</b>
<b>endFor</b>	<b>isKindOf</b>	<b>uses</b>
<b>endFunction</b>	<b>Max_Stack_Usage</b>	<b>var</b>
<b>endHeader</b>	<b>messages</b>	<b>void</b>
<b>endif</b>	<b>method</b>	<b>while</b>
<b>endInterface</b>	<b>methods</b>	<b>word</b>



# About This Document

## Document Revision History

Version numbers are not used to identify revisions to this document. Instead the date and the author's name are used. The document history is:

<u>Date</u>	<u>Author</u>
21 October 2019	Harry H. Porter III <document created>
18 February 2020	Harry H. Porter III <initial version completed>
8 March 2021	Harry H. Porter III
14 December 2023	Harry H. Porter III <current version>

## Permission to Copy

In the spirit of the open-source and free software movements, the author grants permission to freely copy and/or modify this document, with the following requirement:

***You must not alter this section, except to add to the revision history. You must append your date/name to the revision history.***

Any material lifted should be referenced.